

crossdev() — Deviation cross products

Description
Diagnostics

Syntax
Also see

Remarks and examples

Conformability

Description

`crossdev()` makes calculations of the form

$$(X: -x)' (X: -x)$$

$$(X: -x)' (Z: -z)$$

$$(X: -x)' \text{diag}(w) (X: -x)$$

$$(X: -x)' \text{diag}(w) (Z: -z)$$

`crossdev()` is a variation on [M-5] [cross\(\)](#). `crossdev()` mirrors `cross()` in every respect except that it has two additional arguments: `x` and `z`. `x` and `z` record the amount by which `X` and `Z` are to be deviated. `x` and `z` usually contain the (appropriately weighted) column means of `X` and `Z`.

Syntax

real matrix `crossdev(X, x, Z, z)`

real matrix `crossdev(X, x, w, Z, z)`

real matrix `crossdev(X, xc, x, Z, zc, z)`

real matrix `crossdev(X, xc, x, w, Z, zc, z)`

where

`X`: *real matrix* `X`

`xc`: *real scalar* `xc`

`x`: *real rowvector* `x`

`w`: *real vector* `w`

`Z`: *real matrix* `Z`

`zc`: *real scalar* `zc`

`z`: *real rowvector* `z`

Remarks and examples

stata.com

`x` usually contains the same number of rows as `X` but, if `xc` \neq 0, `x` may contain an extra element on the right recording the amount from which the constant 1 should be deviated.

The same applies to `z`: it usually contains the same number of rows as `Z` but, if `zc` \neq 0, `z` may contain an extra element on the right.

▷ Example 1: Linear regression using one view

```

: M = .
: st_view(M, ., "mpg weight foreign", 0)
:
: means = mean(M, 1)
: CP = crossdev(M,means, M,means)
: XX = CP[|2,2 \ .,.]
: Xy = CP[|2,1 \ .,1]
: b = invsym(XX)*Xy
: b = b \ means[1] - means[|2\.] *b

```

Compare this solution with [example 3](#) in [\[M-5\] cross\(\)](#).



▷ Example 2: Linear regression using subviews

```

: M = X = y = .
: st_view(M, ., "mpg weight foreign", 0)
: st_subview(y, M, ., 1)
: st_subview(X, M, ., (2\..))
:
: xmean = mean(X, 1)
: ymean = mean(y, 1)
: XX = crossdev(X,xmean, X,xmean)
: Xy = crossdev(X,xmean, y,ymean)
: b = invsym(XX)*Xy
: b = b \ ymean - xmean*b

```

Compare this solution with [example 4](#) in [\[M-5\] cross\(\)](#).



▷ Example 3: Weighted linear regression

```

: M = X = y = w = .
: st_view(M, ., "w mpg weight foreign", 0)
: st_subview(w, M, ., 1)
: st_subview(y, M, ., 2)
: st_subview(X, M, ., (3\..))
:
: xmean = mean(X, w)
: ymean = mean(y, w)
: XX = crossdev(X,xmean, w, X,xmean)
: Xy = crossdev(X,xmean, w, y,ymean)
: b = invsym(XX)*Xy
: b = b \ ymean - xmean*b

```

Compare this solution with [example 6](#) in [\[M-5\] cross\(\)](#).



▷ Example 4: Variance matrix

```

: X = .
: st_view(X, ., "mpg weight displ", 0)
:
: n      = rows(X)
: means = mean(X, 1)
: xx     = crossdev(X,means, X,means)
: cov    = xx/(n-1)

```

This is exactly what `variance()` does; see [M-5] `mean()`. Compare this solution with [example 12](#) in [M-5] `cross()`.

◀

▷ Example 5: Weighted variance matrix

```

: M = w = X = .
: st_view(M, ., "w mpg weight displ", 0)
: st_subview(w, M, ., 1)
: st_subview(X, M, ., (2\..))
:
: n      = colsum(w)
: means = mean(X, w)
: cov    = crossdev(X,means, w, X,means) :/ (n-1)

```

This is exactly what `variance()` does with weighted data; see [M-5] `mean()`. Compare this solution with [example 14](#) in [M-5] `cross()`.

◀

Conformability

```

crossdev(X, xc, x, w, Z, zc, z):
  X:    n × v1  or  1 × 1,  1 × 1 treated as if n × 1
  xc:   1 × 1
        (optional)
  x:    1 × v1  or  1 × v1 + (xc ≠ 0)
  w:    n × 1   or  1 × n   or  1 × 1 (optional)
  Z:    n × v2
  zc:   1 × 1
        (optional)
  z:    1 × v2  or  1 × v2 + (zc ≠ 0)
  result: (v1 + (xc ≠ 0)) × (v2 + (zc ≠ 0))

```

Diagnostics

`crossdev(X, xc, x, w, Z, zc, z)` omits rows in `X` and `Z` that contain missing values.

Also see

[M-5] `cross()` — Cross products

[M-5] `quadcross()` — Quad-precision cross products

[M-4] **Utility** — Matrix utility functions

[M-4] **Statistical** — Statistical functions

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).